```
/* Lab 1:  basic descriptive info and t-test */


/* part 1 - first example dataset. */
data one;
INPUT y type @@;
datalines;
65 1 64 2
71 1 71 2
57 1 83 2
66 1 69 2
72 1 75 2
71 1 76 2
67 1 69 2
59 1 74 2
75 1 82 2
70 1 79 2
;
run;

/* display the full dataset */
PROC print data=one;
RUN;

/* only show the first 10 observations */
PROC print data=one (obs=10);
RUN;

/*draw a dot plot for the variable "y" */
proc sgplot data=one;
  dot type / response=y stat=mean
             limitstat=stddev numstd=1;
run;

/* draw a vertical or horizontal boxplot*/
proc sgplot data=one;
  vbox y / category=type;
run;

proc sgplot data=one;
  hbox y / category=type;
run;

/* draw a histogram with a normal density curve and a kernel density curve */
proc sgplot data=one;
  histogram y;
  density y;
  density y / type=kernel;
run;

/* side-by-side histogram plot */
proc sgpanel data = one;
  panelby type;
  histogram y;
  density y;
     density y / type=kernel;
run;

/* Within each type group, get basic descriptive info
 for the response variable "y" */

proc univariate data=one;
   class type;
   var y;
run;


/* one sample t-test (e.g., H0=75), with conf. interval included in the results */
proc ttest data=one H0=75 alpha=0.05;
  var y;
run;

/* two-sample t-test, confidence interval for the difference of two group means are included in the result */
proc ttest data=one alpha=0.05;
  class type;
  var y;
run;

/* Within each type, use test statistics to check the normality assumption */
proc univariate data=one NORMALTEST;
   class type;
   var y;
run;



/* part 2: another example: more complicated dataset */
/* HOW TO IMPORT A DATASET INTO SAS */

 PROC IMPORT DATAFILE="/home/u63048916/STAT571B/Labs/hs1.csv"
    DBMS=CSV
    OUT=hs1
    REPLACE;
    GETNAMES=YES;
RUN;


 PROC IMPORT DATAFILE="/home/u63048916/STAT571B/Labs/hs1.xlsx"
    DBMS=XLSX
    OUT=hs1
    REPLACE;
    GETNAMES=YES;
```

```
RUN;



/* take a look at part of data*/
proc print data=hs1 (obs=20);
run;


/*draw dot plot of â??writeâ?? for category variable â??prgtypeâ?? with statistical limits added*/
proc sgplot data=hs1;
  dot prgtype / response=write stat=mean
            limitstat=stddev numstd=1;
run;

/* draw scatter plot with ellipse */
proc sgplot data=hs1;
  scatter x=write y=read;
  ellipse x=write y=read;
run;

/* draw vertical or horizontal boxplot*/
proc sgplot data=hs1;
  vbox write / category=prgtype;
run;

proc sgplot data=hs1;
  hbox write / category=prgtype;
run;

/* draw histogram with a normal density curve, and a kernel density curve */
proc sgplot data=hs1;
  histogram write;
  density write;
  density write / type=kernel;
run;

/* draw bar chart*/
proc sgplot data=hs1;
  yaxis label="score";
  vbar prgtype / response=read;
  vbar prgtype / response=write
                 barwidth=0.5
                 transparency=0.2;
run;


/* get basic descriptive info for two variables "read" and "write "*/
proc univariate data=hs1;
   var read write;
run;

/* check normality for the variable read using test and Q-Q plot*/
proc univariate data=hs1 NORMALTEST;
  class prgtype;
   var read;
QQPLOT read/NORMAL(MU=EST SIGMA=EST COLOR=RED L=1);
run;

/* one sample T- test, conf. interval is included in the results */
proc ttest data=hs1 H0=50 alpha=0.05;
  var write;
run;

/* two -sample t-test, confidence interval for the difference of two group means are included in the result */
proc ttest data=hs1 alpha=0.05;
  class gender;
  var write;
run;


/* paired t-test*/
proc ttest data=hs1 alpha=0.05;
  paired write*read;
run;

/* calculate t critical value */
data crittvals;
input  alpha  df;
tvalue = tinv(1-(alpha/2),df);
datalines;
0.05 18
0.1 18
;

proc print data=crittvals;
```