```
/* Part1: ANOVA for factorial design */
proc import datafile="/home/u63048916/STAT571B/Labs/Lab Data/battery.csv"
    dbms=csv
    out=battery
    replace;
    getnames=yes;
run;

proc print data=battery;
run;

proc glm data=battery;
class mat temp;
model life=mat temp temp*mat;
means mat temp mat*temp;
output out=batnew r=res p=pred;
run;

PROC univariate data=batnew normal;
var res;
qqplot res /normal(MU=0 SIGMA=EST);
run;

proc sgplot;
scatter x=pred y=res;
refline 0;
run;

proc means data=battery;
var life;
by mat temp;
output out=batterymean mean=mn;
run;

proc sgplot data=batterymean;
 series x=temp y=mn/group=mat;
run;

/* multiple comparison */

proc glm data=battery;
class mat temp;
model life=mat temp mat*temp;
means mat|temp /tukey lines;
output out=batnew1 r=res p=pred;
run;

proc glm data=battery;
class mat temp;
model life=mat temp mat*temp;
lsmeans mat|temp/tdiff adjust=tukey;
output out=batnew2 r=res p=pred;
run;



/* Part2: single replication (i.e., one observation) factorial design */
data one;
input Row   Column  value;
datalines;
1 1 36
2 1 18
3 1 30
1 2 39
2 2 20
3 2 37
1 3 36
2 3 22
3 3 33
1 4 32
2 4 20
3 4 34
;
run;

proc glm data=one;
class Row Column;
model value = Row Column;
output out=diag r=res p=pred;
Run;

proc univariate data=diag normal;
var res;
qqplot res / normal (mu=est sigma=est);
Run;

title 'residual plot: res vs predicted value ';
proc sgplot data=diag;
scatter x=pred y=res;
refline 0;
run;

/* check additivity for one observation factorial design - same as RCBD*/
data two;
set diag;
q=pred*pred;
run;

proc glm data=two;
class Row Column;
model value = Row Column q/ss3;
run;
```

```
/* Part3: blocking in factorial design */

proc import datafile="/home/u63048916/STAT571B/Labs/Lab Data/battery_block.csv"
    dbms=csv
    out=bb
    replace;
    getnames=yes;
run;

proc print data=bb;
run;

proc glm data=bb;
class mat temp operator;
model life=operator temp|mat;
means operator mat temp mat*temp;
output out=new1 r=res p=pred;
run;

/* then check model adequacy */
```