

```

/* Lab-4A: RCBD design */

data wash;
input stain soap y @@;
datalines;
1 1 45 1 2 47 1 3 48 1 4 42 2 1 43 2 2 46 2 3 50 2 4 37 3 1 51 3 2
52 3 3 55 3 4 49
;
run;
proc print data=wash;
run;

proc glm data=wash;
class stain soap;
model y = soap stain;
output out=diag r=res p=pred;
Run;

/* check normality */
proc univariate data=diag normal;
var res;
qqplot res / normal (mu=est sigma=est);
Run;

/* check outliers */
data outlier;
set diag;
stdres=res/1.771691;
run;

proc print data=outlier;
run;

/* check constant variance using graph*/
title 'residual plot: res vs predicted value ';
proc sgplot data=diag;
scatter x=pred y=res;
refline 0;
run;

title 'residual plot: res vs soap ';
proc sgplot data=diag;
scatter x=soap y=res;
refline 0;
run;

title 'residual plot: res vs stain ';
proc sgplot data=diag;
scatter x=stain y=res;
refline 0;
run;

/* check additivity for RCBD design (with one replicate)*/
data two;
set diag;
q=pred*pred;
run;

proc glm data=two;
class soap stain;
model y = soap stain q/ss3;
run;

/* pairwise compsisons */
proc glm data=wash;
class stain soap;
model y = soap stain;
means soap / alpha=0.05 tukey lines;
means stain / alpha=0.05 tukey lines;
output out=diag r=res p=pred;
Run;

/* Another example: how to handle missing values */
data wash_missing;
input stain soap y @@;
datalines;
1 1 45 1 2 47 1 3 48 1 4 42
2 1 43 2 2 46 2 3 50 2 4 .
3 1 51 3 2 52 3 3 55 3 4 49
;

proc print data=wash_missing;
run;

proc glm data=wash_missing;
class stain soap;
model y = soap stain;
output out=diag r=res p=pred;
run;

title 'wrong way - use the regular way - means - for post-ANOVA comparison with missing data ';
proc glm data=wash_missing;
class stain soap;
model y = soap stain;
means soap / alpha=0.05 tukey lines;
output out=diag1 r=res p=pred;
run;

title 'use the correct way - lsmeans - for post-ANOVA comparison with missing data';
proc glm data=wash_missing;
class stain soap;

```

```
model y = soap stain;  
lsmeans soap / alpha=0.05 adjust=tukey;  
output out=diag2 r=res p=pred;  
run;
```