```
/* Lab 6: 2^k factorial design */

/* part 1: 2 factor factorial design. */

data one;
Input A B resp;
AB=A*B;
datalines;
-1 -1 28
-1 -1 25
-1 -1 27
1 -1 36
1 -1 32
1 -1 32
-1 1 18
-1 1 19
-1 1 23
1 1 31
1 1 30
1 1 29
;

proc print data=one;
run;

proc glm data=one;
class A B;
model resp=A B A*B;   /* use A*B if you want to generate an interaction plot */
means A|B;
output out=diag r=res p=pred;
run;

proc glm data=one;
class A B AB;
model resp=A B AB;  /* Use AB if you want to estimate the interaction effect */
means A|B;
estimate "A" A -1 1;
estimate "B" B -1 1;
estimate "AB" AB -1 1;
output out=diag r=res p=pred;
run;

proc sgplot data=diag;
scatter y=res x=pred;
refline 0;
run;

/* wrong model */
proc glm data=one;
class A B;
model resp=A|B;
estimate "A" A -1 1;
estimate "B" B -1 1;
estimate "AB" AB -1 1;
output out=diag r=res p=pred;
run;

/* part 2: regression */

data two;
Input x1 x2 resp;
x1x2=x1*x2;  /* need to define the interaction term in the data step, for use in proc reg later */
datalines;
-1 -1 28
-1 -1 25
-1 -1 27
1 -1 36
1 -1 32
1 -1 32
-1 1 18
-1 1 19
-1 1 23
1 1 31
1 1 30
1 1 29
;

proc reg data=two;
model resp=x1 x2 x1x2;
run;


 /* wrong model */
proc reg data=two;
model resp=x1 x2 x1*x2;
run;


/* part 3: single replicate */

data filter;
input A B C D y;
datalines;
-1 -1 -1 -1 45
 1 -1 -1 -1 71
-1  1 -1 -1 48
 1  1 -1 -1 65
-1 -1  1 -1 68
 1 -1  1 -1 60
-1  1  1 -1 80
 1  1  1 -1 65
-1 -1 -1  1 43
```

```
 1 -1 -1  1 100
-1  1 -1  1 45
 1  1 -1  1 104
-1 -1  1  1 75
 1 -1  1  1 86
-1  1  1  1 70
 1  1  1  1 96
;
proc print data=filter;
run;


data inter;                              /* Define Interaction Terms */
 set filter;
 AB=A*B; AC=A*C; AD=A*D; BC=B*C; BD=B*D; CD=C*D; ABC=AB*C; ABD=AB*D;
 ACD=AC*D; BCD=BC*D; ABCD=ABC*D;
run;

proc print data=inter;
run;


proc glm data=inter;                    /* GLM Proc to Obtain Effects */
 class A B C D AB AC AD BC BD CD ABC ABD ACD BCD ABCD;
 model y=A B C D AB AC AD BC BD CD ABC ABD ACD BCD ABCD;
 estimate 'A' A -1 1;
 estimate "B" B -1 1;
 estimate "C" C -1 1;
 estimate "D" D -1 1;
 estimate "AB" AB -1 1;
estimate 'AC' AC -1 1;
estimate 'AD' AD -1 1;
estimate 'BC' BC -1 1;
estimate 'BD' BD -1 1;
estimate 'CD' CD -1 1;
estimate 'ABC' ABC -1 1;
estimate 'ABD' ABD -1 1;
estimate 'ACD' ACD -1 1;
estimate 'BCD' BCD -1 1;
estimate 'ABCD' ABCD -1 1;
 run;


proc reg data=inter outest=effects;    /* REG Proc to Obtain Effects */
 model y=A B C D AB AC AD BC BD CD ABC ABD ACD BCD ABCD;
 run;

 proc print data=effects;
 run;

data effect2; set effects;
 drop y intercept _RMSE_;
 run;

proc transpose data=effect2 out=effect3;
run;

proc print data=effect3;
 run;

data effect4; set effect3; effect=col1*2;
run;

proc sort data=effect4; by effect;
run;

/* print out the effects */
proc print data=effect4;
run;

/* draw plot for selecting significant effects */
proc rank data=effect4 out=effect5 normal=blom;
 var effect;
ranks neff;
 run;

proc print data=effect5;
run;

proc sgplot data=effect5;
scatter x=neff y=effect/datalabel=_NAME_;
xaxis label='Normal Scores';
run;


 /* keep A, C and D and their interactions */

proc glm data=filter;
class A C D;
model y=A|C|D;
run;

proc reg data=inter;
model y=A C D AC AD;
output out=outres r=res p=pred;
run;


/* part 4: best setting selection */

/* for D=1 the fitted regression model is y=77.37+19.12*x1 +4.94*x3 -9.06*x1*x3 */
data one;
do x1 = -1 to 1 by 0.1;
```

```
do x3 = -1 to 1 by 0.1;
y=77.37+19.12*x1 +4.94*x3-9.06*x1*x3 ;
output;
end;
end;
proc gcontour data=one;
plot x3*x1=y;
run;


/* for D=-1 the fitted regression model is y=62.75+2.5*x1 +4.94*x3 -9.06*x1*x3*/
data one;
do x1 = -1 to 1 by 0.1;
do x3 = -1 to 1 by 0.1;
y=62.75+2.5*x1 +4.94*x3-9.06*x1*x3 ;
output;
end;
end;
proc gcontour data=one;
plot x3*x1=y;
run;

/* alternative way to find the optimal level */
proc glm data=filter;
class A C D;
model y=A|C|D;
means A*C*D;
run;
```